# $Id$
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: James Hunt <lazy at brandnewmath dot com>

pkgname=perl-xml-sax-base
pkgver=1.09
pkgrel=2
pkgdesc="Base class SAX Drivers and Filters"
arch=('any')
url="http://search.cpan.org/dist/XML-SAX-Base"
license=('GPL' 'PerlArtistic')
depends=('perl')
options=('!emptydirs')
source=("http://www.cpan.org/authors/id/G/GR/GRANTM/XML-SAX-Base-$pkgver.tar.gz")
md5sums=('ec347a14065dd7aec7d9fb181b2d7946')

build() {
  cd  $srcdir/XML-SAX-Base-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd  $srcdir/XML-SAX-Base-$pkgver
  make test
}

package() {
  cd  $srcdir/XML-SAX-Base-$pkgver
  make install DESTDIR=$pkgdir

  find $pkgdir -name '.packlist' -delete
  find $pkgdir -name '*.pod' -delete
}
